Formalized q-series

Seewoo Lee, (was) UC Berkeley

July 16, 2026. Thematic Program on AI and Mathematics, KIAS

Goal

We formalize q-series in Lean, from the basic definitions through Bailey's lemma, q-hypergeometric series, the Jacobi Triple Product identity, Euler's Pentagonal Number Theorem, and the Rogers–Ramanujan identities.

We do so in the most general setting available to us.

This talk is based on joint work with Kenny Lau and Ken Ono at Axiom Math  .

q-series: What, why, where, and when

q

What is q?

q

What is q?

- $q = p^n$ for some prime p; $\mathbb{F}_q$ is the finite field with q elements.

q

What is q?

- $q = p^n$ for some prime p; $\mathbb{F}_q$ is the finite field with q elements.
- Deformation or **q**uantization. ($q \to 1$ is "classical").

$$n \rightsquigarrow \frac{q^n - 1}{q - 1}, \quad \mathcal{U}(\mathfrak{sl}_2) \rightsquigarrow \mathcal{U}_q(\mathfrak{sl}_2)$$

q

What is q?

- $q = p^n$ for some prime p; $\mathbb{F}_q$ is the finite field with q elements.
- Deformation or **q**uantization. ($q \to 1$ is "classical").

$$n \rightsquigarrow \frac{q^n - 1}{q - 1}, \quad \mathcal{U}(\mathfrak{sl}_2) \rightsquigarrow \mathcal{U}_q(\mathfrak{sl}_2)$$

- Formulae written in Ramanujan's notebooks



The main identities

We are mostly interested in the third point: identities involving q-series. There are tons of q-series in the world, but today our main focus is on the following identities:

Theorem (Euler's Pentagonal Number Theorem, PNT)

$$\sum_{n=-\infty}^{\infty} (-1)^n q^{\frac{3n^2+n}{2}} = \prod_{m=1}^{\infty} (1-q^m)$$

Theorem (Jacobi Triple Product identity, JTP)

$$\sum_{n=-\infty}^{\infty} z^n q^{n^2} = \prod_{m=0}^{\infty} (1-q^{2m+2})(1+zq^{2m+1})(1+z^{-1}q^{2m+1})$$

The main identities

We also focus on these identities:

Theorem (Rogers–Ramanujan identities, RR)

$$G(q) := \sum_{n=0}^{\infty} \frac{q^{n^2}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+1})(1-q^{5n+4})},$$

$$H(q) := \sum_{n=0}^{\infty} \frac{q^{n^2+n}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+2})(1-q^{5n+3})}.$$

Why they matter

The above identities connect several areas of mathematics (and physics):

- (Jacobi) Let $r_2(n)$ be the number of representations of n as a sum of two squares. Then $r_2(n) = 4\sum_{2\nmid d\mid n}(-1)^{\frac{d-1}{2}}$.
- The Weil denominator formula for the affine Kac–Moody algebra $\widehat{\mathfrak{sl}_2}$.
- (Berndt–Chan–Zhang, Cais–Conrad) Let $f(z) = F(q) := q^{1/5}\frac{H(q)}{G(q)}$, where $q = e^{2\pi iz}$. If $z \in \mathbb{H}$ is a CM point, then $f(z)$ is an algebraic unit that can be expressed in terms of radicals over $\mathbb{Q}$.
- (Baxter) $F(q)$ gives a solution for the hard hexagon model in statistical mechanics.
- (Macdonald) $G(q^{-1}) = \sum_{n\geq 0}|\mathrm{GL}_n(\mathbb{F}_q)|^{-1}$ for q a prime power.

Where do these identities hold?

$$\sum_{n=0}^{\infty} \frac{q^{n^2}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+1})(1-q^{5n+4})},$$

$$\sum_{n=0}^{\infty} \frac{q^{n^2+n}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+2})(1-q^{5n+3})}.$$

What are these identities, exactly? *Where* is q?

Where do these identities hold?

$$\sum_{n=0}^{\infty} \frac{q^{n^2}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+1})(1-q^{5n+4})},$$
$$\sum_{n=0}^{\infty} \frac{q^{n^2+n}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+2})(1-q^{5n+3})}.$$

What are these identities, exactly? *Where* is q? In the literature, they usually mean:

- identities of analytic functions for $q \in \mathbb{C}$ with $|q| < 1$;
- identities of formal power series in $\mathbb{C}[\![q]\!]$ (or in $\mathbb{Z}[\![q]\!]$).

Where do these identities hold?

$$\sum_{n=0}^{\infty} \frac{q^{n^2}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+1})(1-q^{5n+4})},$$
$$\sum_{n=0}^{\infty} \frac{q^{n^2+n}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+2})(1-q^{5n+3})}.$$

What are these identities, exactly? *Where* is q? In the literature, they usually mean:

- identities of analytic functions for $q \in \mathbb{C}$ with $|q| < 1$;
- identities of formal power series in $\mathbb{C}[\![q]\!]$ (or in $\mathbb{Z}[\![q]\!]$).

Can we go further?

Where do these identities hold?

$$\sum_{n=0}^{\infty} \frac{q^{n^2}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+1})(1-q^{5n+4})},$$
$$\sum_{n=0}^{\infty} \frac{q^{n^2+n}}{(1-q)(1-q^2)\cdots(1-q^n)} = \frac{1}{\prod_{n=0}^{\infty}(1-q^{5n+2})(1-q^{5n+3})}.$$

What are these identities, exactly? *Where* is q? In the literature, they usually mean:

- identities of analytic functions for $q \in \mathbb{C}$ with $|q| < 1$;
- identities of formal power series in $\mathbb{C}[\![q]\!]$ (or in $\mathbb{Z}[\![q]\!]$).

Can we go further? Spoiler: Yes!

When were they first formalized?

We are *not* the first to formalize these identities. Earlier formalizations include:

- Weiyi Wang (in Lean): PNT
- Fabian Glöckle et al. (in Lean, using RepoProver): PNT, JTP
- Manuel Eberl (in Isabelle/HOL): PNT, JTP, RR

When were they first formalized?

We are *not* the first to formalize these identities. Earlier formalizations include:

- Weiyi Wang (in Lean): PNT — power series over R, with R Hausdorff + convergence hypotheses
- Fabian Glöckle et al. (in Lean, using RepoProver): PNT, JTP — power series over $\mathbb{Z}[z^{\pm 1}]$
- Manuel Eberl (in Isabelle/HOL): PNT, JTP, RR — functions on $\mathbb{C}$

We work over *strongly non-archimedean rings*, which subsume all of the above settings.[1]

[1] For $\mathbb{Z}[z^{\pm 1}][\![q]\!]$, we use the q-adic topology, with the discrete topology on $\mathbb{Z}[z^{\pm 1}]$. Eberl's work can be viewed as identities in $\mathbb{C}[\![q]\!]$, which is strongly non-archimedean.

And more!

- q-Pochhammer symbols, q-integer, q-factorial, q-binomial coefficients, etc.
- A convergence theory for infinite products in strongly non-archimedean rings
- q-hypergeometric transformations and Bailey's lemma

Basics of q-series

Definitions

The basic definitions are:

- q-Pochhammer symbol

$$(a;q)_n = \prod_{i=0}^{n-1}(1-aq^i), \quad (a;q)_\infty = \prod_{i=0}^{\infty}(1-aq^i)$$

- q-integer

$$[n]_q = 1 + q + \cdots + q^{n-1}$$

- q-factorial

$$[n]_q! = [1]_q[2]_q \cdots [n]_q$$

- q-binomial coefficient

$$\begin{bmatrix} n \\ k \end{bmatrix}_q = \frac{[n]_q!}{[k]_q![n-k]_q!}$$

q-Pochhammer symbol

The finite q-Pochhammer symbol can be formalized directly:

```
def qPochhammer {R : Type*} [CommRing R] (a q : R) (n : ℕ) : R :=
  ∏ i ∈ range n, (1 - a * q ^ i)
```

q-Pochhammer symbol

The finite q-Pochhammer symbol can be formalized directly:

```
def qPochhammer {R : Type*} [CommRing R] (a q : R) (n : ℕ) : R :=
  ∏ i ∈ range n, (1 - a * q ^ i)
```

The infinite q-Pochhammer symbol requires an infinite product, hence a topology on the ring.

```
noncomputable def qPochhammerInf {R : Type*} [TopologicalSpace R] [CommRing R] (a q : R) : R :=
  ∏'[conditional] i, (1 - a * q ^ i)
```

Here, the `conditional` filter gives the usual limit $(a;q)_\infty = \lim_{n\to\infty}(a;q)_n$.

q-integer and q-factorial

q-integers and q-factorials only use addition and multiplication:

```
def qInt {R : Type*} [CommSemiring R] (q : R) (n : ℕ) : R :=
  ∑ i ∈ range n, q ^ i

def qFactorial {R : Type*} [CommSemiring R] (q : R) (n : ℕ) : R :=
  ∏ i ∈ range n, qInt q (i + 1)
```

Since subtraction is unnecessary, `CommSemiring R` suffices in place of `CommRing R`.

q-binomial coefficient

How about q-binomial coefficients? We don't like division.

q-binomial coefficient

How about q-binomial coefficients? We don't like division.

Instead, we can ask how the usual binomial coefficient `Nat.choose` is defined.

q-binomial coefficient

How about q-binomial coefficients? We don't like division.

Instead, we can ask how the usual binomial coefficient `Nat.choose` is defined.

```
def choose : ℕ → ℕ → ℕ
  | _, 0 => 1
  | 0, _ + 1 => 0
  | n + 1, k + 1 => choose n k + choose n (k + 1)
```

Aha!

q-binomial coefficient

How about q-binomial coefficients? We don't like division.

Instead, we can ask how the usual binomial coefficient `Nat.choose` is defined.

```
def choose : ℕ → ℕ → ℕ
  | _, 0 => 1
  | 0, _ + 1 => 0
  | n + 1, k + 1 => choose n k + choose n (k + 1)
```

Aha!

```
def qChoose {R : Type*} [CommSemiring R] (q : R) : ℕ → ℕ → R
  | _, 0 => 1
  | 0, _ + 1 => 0
  | n + 1, k + 1 => qChoose q n k + q ^ (k + 1) * qChoose q n (k + 1)
```

The expected factorial formula and algebraic identities are then proved from this recursive definition.

Strongly non-archimedean rings

What replaces the condition $|q| < 1$?

Classical q-series are usually developed over $\mathbb{C}$, where $|q| < 1$ controls convergence.

We want the same identities to apply to formal power series, p-adic rings, and more general topological rings.

The most natural generalization is to consider *topologically nilpotent* element q in a topological ring R. But is this enough?

Question

Which algebraic and topological hypotheses recover the convergence properties used in classical proofs?

Non-archimedean ring

Definition

A **non-archimedean ring** is a topological ring R such that the set of open **additive** subgroups of R forms a neighborhood basis of 0. In other words, for every open neighborhood U of 0, there exists an open additive subgroup V of R such that $V \subseteq U$.

The following rings are non-archimedean:

- The power series ring $R[\![x]\!]$ with the x-adic topology
- The p-adic integers $\mathbb{Z}_p$
- Tate algebra
- Product of non-archimedean rings

Non-archimedean ring?

Question

Is a non-archimedean ring enough for q-theory?

Non-archimedean ring?

Question

Is a non-archimedean ring enough for q-theory?

Answer

No! **It does not guarantee that multiplication works well enough.**

Non-archimedean ring?

Question

Is a non-archimedean ring enough for q-theory?

Answer

No! **It does not guarantee that multiplication works well enough.**

In particular, the product $\prod_{n\geq 0}(1-aq^n)$ may not converge even if q is topologically nilpotent, if we only assume that R is non-archimedean.

Take

$$R = \left\{ x = (x_n) \in \mathbb{Q}_p^{\mathbb{N}} : \forall N \in \mathbb{N}_+, \|x\|_N := \sup_n \left(p^{-n^3/N} \|x_n\|_p \right) < \infty \right\}$$

and $a = (p^{-n^2})_n = (p^{-1}, p^{-4}, p^{-9}, \ldots)$, $q = (p)_n = (p, p, p, \ldots)$. Then $a, q \in R$ and q is topologically nilpotent, but $\lim_{n\to\infty} \prod_{k=0}^{n}(1-aq^k)$ does not exist in R.

We need something stronger

Definition

A **strongly non-archimedean ring** is a topological ring whose neighborhoods of 0 admit a basis of open *subrngs*: additive subgroups **that are also closed under multiplication**.

```
class StrongNonarchimedeanRing (R) [Ring R] [TopologicalSpace R] :
    Prop extends NonarchimedeanRing R where
  exists_mul_subset_self (U : Set R) (hU : U ∈ 𝓝 0) :
    ∃ V : Set R, 0 ∈ V ∧ IsOpen V ∧ V ⊆ U ∧ V * V ⊆ V
```

Many familiar rings are strong

- Any normed ring equipped with an ultrametric norm
- Any topological ring R with an R-linear topology (i.e., the open ideals of R form a neighborhood basis of 0)
- Multivariate Laurent series[2]

```
instance {F} [NormedRing F] [IsUltrametricDist F] : StrongNonarchimedeanRing F

instance IsLinearTopology.instStrongNonarchimedeanRing {R} [Ring R] [TopologicalSpace R]
  [IsTopologicalRing R] [IsLinearTopology R R] : StrongNonarchimedeanRing R

instance {σ R} [Ring R] : StrongNonarchimedeanRing (MvLaurentSeries σ R)
```

[2]Defined as Hahn series with the order determined by the total degree.

Strongness makes infinite products converge

Theorem

Let R be a strongly non-archimedean ring, and let $(a_n)_{n\in\mathbb{N}}$ be a sequence in R. If $\lim_{n\to\infty} a_n = 0$, then $\lim_{n\to\infty} \prod_{i=0}^{n-1} a_i = 0$. In particular, $\lim_{n\to\infty} a^n q^{\binom{n}{2}} = 0$ if q is topologically nilpotent. If R is also complete and $\lim_{n\to\infty} a_n = 0$, then $\prod_n (1 + a_n)$ converges unconditionally.

```
theorem prod_tendsto_zero {R} [CommRing R] [TopologicalSpace R] [StrongNonarchimedeanRing R]
  (a : ℕ → R) (ha : Tendsto a atTop (𝓝 0)) : Tendsto (∏ i ∈ range ·, a i) atTop (𝓝 0) :=

theorem multipliable_one_add_of_tendsto_zero {R} [CommRing R] [UniformSpace R]
  [CompleteSpace R] [StrongNonarchimedeanRing R] [IsUniformAddGroup R]
  {a : ℕ → R} (ha : Tendsto a atTop (𝓝 0)) : Multipliable (1 + a ·) :=
```

Strongness gives infinite q-Pochhammer products

In particular, we can finally prove that $(a;q)_\infty$ converges.

Theorem

Let R be a complete strongly non-archimedean ring, and $a, q \in R$ with q topologically nilpotent. Then the infinite q-Pochhammer symbol $(a;q)_\infty$ exists unconditionally.

```
theorem multipliable_one_sub_mul_pow {R} [CommRing R] [UniformSpace R] [CompleteSpace R]
  [StrongNonarchimedeanRing R] [IsUniformAddGroup R]
  {a q : R} (hq : IsTopologicallyNilpotent q) :
  Multipliable (1 - a * q ^ ·) :=
```

Infinite q-products are functorial

Theorem

Let $f\colon R \to S$ be a continuous ring homomorphism, where R is strongly non-archimedean and S is a Hausdorff ring. For $a, q \in R$ with q topologically nilpotent, we have $f((a;q)_\infty) = (f(a); f(q))_\infty$.

```
theorem map_qPochhammerInf {R S F} [CommRing R] [UniformSpace R]
  [StrongNonarchimedeanRing R] [CompleteSpace R] [IsUniformAddGroup R]
  [CommRing S] [TopologicalSpace S] [T2Space S]
  [FunLike F R S] [RingHomClass F R S] (f : F) (hf : Continuous f)
  (a : R) {q : R} (hq : IsTopologicallyNilpotent q) :
  f (a; q)_∞ = (f a; f q)_∞ :=
```

Note that we don't assume that S is strongly non-archimedean.

q-hypergeometric series

q-hypergeometric series

A q-hypergeometric series is defined by

$$ {}_r\phi_s\left(a_0,\dots,a_{r-1};b_0,\dots,b_{s-1};q,t\right)=\sum_{n=0}^{\infty}\frac{(a_0;q)_n(a_1;q)_n\cdots(a_{r-1};q)_n}{(b_0;q)_n(b_1;q)_n\cdots(b_{s-1};q)_n}\left((-1)^n q^{\frac{n(n-1)}{2}}\right)^{s+1-r}t^n $$

```
noncomputable def qHypergeometricInner {R} [CommRing R]
    (q : R) {r s : ℕ} (a : Fin r → R) (b : Fin s → R) (t : R) (e n : ℕ) :=
  (∏ i, (a i)_n) * bInv (q)_n * (∏ j, bInv (b j)_n) * ((-1) ^ n * q ^ n.choose 2) ^ e * t ^ n

noncomputable def qHypergeometric {R} [CommRing R] [TopologicalSpace R]
    (q : R) {r s : ℕ} (a : Fin r → R) (b : Fin s → R) (t : R) : R :=
  ∑' n, qHypergeometricInner q a b t (s + 1 - r) n
```

The `bInv` API returns an actual inverse when an element is invertible and 1 (a junk value) otherwise.

Identities for q-hypergeometric series

We formalized many identities for q-hypergeometric series, including:

- q-binomial series identity (${}_1\phi_0$)
- Heine's transformation (${}_2\phi_1$)
- q-Gauss theorem (${}_2\phi_1$)
- Rogers' identities (${}_2\phi_1$)
- q-Pfaff–Saalschütz theorem (${}_3\phi_2$)

q-binomial series identity

The q-binomial series identity is given by

$$ {}_1\phi_0(a; -; q, z) = \sum_{n=0}^{\infty} \frac{(a; q)_n}{(q; q)_n} z^n = \frac{(az; q)_\infty}{(z; q)_\infty} $$

```
theorem qPochhammerInf_div_qPochhammerInf_eq_tsum {R} [CommRing R] [UniformSpace R]
  [IsUniformAddGroup R] [StrongNonarchimedeanRing R] [CompleteSpace R] [T2Space R]
  {a q z : R} (hz : IsTopologicallyNilpotent z := by simp)
  (hq : IsTopologicallyNilpotent q := by simp) :
  ((a * z); q)_∞ * bInv ((z; q)_∞) = ∑' n, (a; q)_n * bInv (q; q)_n * z ^ n :=
```

q-binomial series identity over $\mathbb{C}$

$$ {}_1\phi_0(a;-;q,z) = \sum_{n=0}^{\infty} \frac{(a;q)_n}{(q;q)_n} z^n = \frac{(az;q)_\infty}{(z;q)_\infty} $$

Over $\mathbb{C}$, the standard proof uses analytic continuation. Assume $|q| < 1$ and $|z| < 1$. Each side, regarded as a function $F(z)$, satisfies the q-difference equation

$$ (1-z)F(z) = (1-az)F(qz), \qquad F(0) = 1. $$

Both sides are holomorphic on the unit disc, so uniqueness gives the identity.

q-binomial series identity over a strongly non-archimedean ring

In our setup, we take a different approach. The infinite q-binomial theorem (which uses strongly non-archimedeanness) gives

$$(az;q)_\infty = \sum_{k=0}^{\infty} \frac{(-az)^k q^{\binom{k}{2}}}{(q;q)_k}, \qquad \frac{1}{(z;q)_\infty} = \sum_{m=0}^{\infty} \frac{z^m}{(q;q)_m}.$$

Multiplying and grouping the double sum by $n = k + m$ gives

$$\frac{(az;q)_\infty}{(z;q)_\infty} = \sum_{n=0}^{\infty} \frac{z^n}{(q;q)_n} \sum_{k=0}^{n} \begin{bmatrix} n \\ k \end{bmatrix}_q (-a)^k q^{\binom{k}{2}}.$$

The inner sum is $(a;q)_n$ by the finite q-binomial theorem, proving the identity.

Bailey's lemma

Bailey pair

Bailey's lemma gives a way to generate new q-series identities from old ones.

Two sequences (α_n, β_n) form a **Bailey pair relative to** a if, for every $n \geq 0$,

$$\beta_n = \sum_{r=0}^{n} \frac{\alpha_r}{(q;q)_{n-r}(aq;q)_{n+r}}.$$

```
noncomputable def Bailey.IsBaileyPair {R} [CommRing R] (a q : R) (α β : ℕ → R) : Prop :=
  ∀ (n : ℕ), β n = ∑ r ∈ range (n + 1),
    α r * bInv (qPochhammer q q (n - r)) * bInv (qPochhammer (a * q) q (n + r))
```

Bailey's lemma

The original form of Bailey's lemma is the following:

Lemma (Bailey)

Suppose (α_n, β_n) *is a Bailey pair relative to* a*, and put* $c = aq/(\rho_1\rho_2)$*. Define*

$$\alpha'_n = \frac{(\rho_1;q)_n(\rho_2;q)_n}{(aq/\rho_1;q)_n(aq/\rho_2;q)_n}\, c^n\alpha_n,$$

$$\beta'_n = \sum_{j=0}^{n} \frac{(\rho_1;q)_j(\rho_2;q)_j(c;q)_{n-j}}{(aq/\rho_1;q)_n(aq/\rho_2;q)_n(q;q)_{n-j}}\, c^j\beta_j.$$

Then (α'_n, β'_n) *is again a Bailey pair relative to* a*.*

The parameters ρ_1 and ρ_2 are freely chosen invertible elements. Applying this lemma repeatedly gives a *Bailey chain*, which produces (infinitely) many q-series identities.

The original form of Bailey's lemma in Lean

```
theorem Bailey.qBaileyLemma {R} [CommRing R] (a q : R)
    (α β α' β' : ℕ → R) (h : IsBaileyPair a q α β)
    (ρ₁ ρ₂ : R) (hρ₁ : IsUnit ρ₁) (hρ₂ : IsUnit ρ₂) (hq : IsUnit q)
    (hqpoc : ∀ k, IsUnit (q)_k) (haqpoc : ∀ k, IsUnit (a * q)_k)
    (haqρ₁ : ∀ k, IsUnit (a * q * bInv ρ₁)_k) (haqρ₂ : ∀ k, IsUnit (a * q * bInv ρ₂)_k)
    (hα' : ∀ n, α' n = (ρ₁)_n * (ρ₂)_n *
      (a * q * bInv (ρ₁ * ρ₂)) ^ n * bInv (a * q * bInv ρ₁)_n * bInv (a * q * bInv ρ₂)_n * α n)
    (hβ' : ∀ n, β' n = ∑ j ∈ range (n + 1),
      (ρ₁)_j * (ρ₂)_j * (a * q * bInv (ρ₁ * ρ₂))_(n - j) *
        (a * q * bInv (ρ₁ * ρ₂)) ^ j * bInv (a * q * bInv ρ₁)_n *
          bInv (a * q * bInv ρ₂)_n * bInv ((q)_(n - j)) * β j) :
    IsBaileyPair a q α' β' :=
```

We assume that the q-Pochhammer symbols in the denominators are invertible.

First limiting version: $\rho_1, \rho_2 \to \infty$

"Sending $\rho_1, \rho_2 \to \infty$" gives a simplified Bailey pair

$$\alpha'_n = a^n q^{n^2} \alpha_n, \qquad \beta'_n = \sum_{j=0}^{n} \frac{a^j q^{j^2}}{(q;q)_{n-j}} \beta_j.$$

But it is not easy to handle such a limit in a general ring. Instead, we prove the result directly using another finite q-series identity. In particular, it does not require any topology on the ring.

```
theorem Bailey.qBaileyLemma_limit {R} [CommRing R] (a q : R)
    (α β α' β' : ℕ → R) (h : IsBaileyPair a q α β)
    (hqpoc : ∀ k, IsUnit (q)_k) (haqpoc : ∀ k, IsUnit (a * q)_k)
    (hα' : ∀ n, α' n = a ^ n * q ^ (n ^ 2) * α n)
    (hβ' : ∀ n, β' n = ∑ j ∈ range (n + 1), a ^ j * q ^ (j ^ 2) * bInv (q)_(n - j) * β j) :
    IsBaileyPair a q α' β' :=
```

Second limiting version: $\rho_1, \rho_2 \to \infty$, $n \to \infty$

Taking the additional limit "$n \to \infty$" gives the following limiting version of Bailey's lemma:

$$\sum_{j=0}^{\infty} a^j q^{j^2} \beta_j = \frac{1}{(aq;q)_\infty} \sum_{r=0}^{\infty} a^r q^{r^2} \alpha_r.$$

```
theorem Bailey.qBaileyLemma_limit' {R} [CommRing R] (a q : R)
    [UniformSpace R] [IsUniformAddGroup R] [CompleteSpace R] [T2Space R] [StrongNonarchimedeanRing R]
    (α β : ℕ → R) (h : IsBaileyPair a q α β)
    (hq : IsTopologicallyNilpotent q) (haq : IsTopologicallyNilpotent (a * q))
    (hα : Tendsto (fun r ↦ a ^ r * q ^ (r ^ 2) * α r) atTop (nhds 0)) :
    ∑' j, a ^ j * q ^ (j ^ 2) * β j =
      bInv (a * q)_∞ * ∑' r, a ^ r * q ^ (r ^ 2) * α r :=
```

The proof uses the infinite q-binomial identity, and we assume that R is strongly non-archimedean.

Finally, JTP, PNT, and RR

Jacobi Triple Product identity

We can rewrite JTP as

$$(q;q)_\infty (a;q)_\infty (a^{-1}q;q)_\infty = \sum_{n\in\mathbb{Z}} (-a)^n q^{\binom{n}{2}}.$$

Two issues arise:

- The formula requires a to be invertible, but our proof uses only the combination $a^{-1}q$.
- We want to apply it to $(q^5;q^5)_\infty (q^2;q^5)_\infty (q^3;q^5)_\infty$ over $\mathbb{Z}[\![q]\!]$ to prove the Rogers–Ramanujan identities. But q^2 is not invertible in $\mathbb{Z}[\![q]\!]$.

Jacobi Triple Product identity

Instead, we formalize the following version of JTP: for $b, c \in R$ with $bc = q$,

$$(q;q)_\infty (b;q)_\infty (c;q)_\infty = \sum_{n \in \mathbb{Z}} e_{-b,-c}(n) q^{\binom{|n|}{2}}, \qquad e_{b,c}(n) = \begin{cases} b^n & n \geq 0 \\ c^{-n} & n < 0 \end{cases}$$

```
def abPow {R} [Pow R ℕ] (a b : R) (n : ℤ) : R := match n with
  | (n : ℕ) => a ^ n
  | Int.negSucc n => b ^ (n + 1)

theorem jacobi_triple_product' {R} [CommRing R] [UniformSpace R]
  [IsUniformAddGroup R] [CompleteSpace R] [StrongNonarchimedeanRing R] [T2Space R]
  {a b q : R} (hq : IsTopologicallyNilpotent q) (hab : a * b = q) :
  (q; q)_∞ * (a; q)_∞ * (b; q)_∞ = ∑' n : ℤ, abPow (-a) (-b) n * q ^ n.natAbs.choose 2 :=
```

There is no inverse! We also have two variants: one with inverses and one over $\mathbb{Z}[\![q]\!]$.

Euler's Pentagonal Number Theorem

Using JTP, we can prove PNT by specializing it to $(a, b, q) := (q, q^2, q^3)$. If q is topologically nilpotent in a complete non-archimedean ring, then

$$(q; q)_\infty = \sum_{k \in \mathbb{Z}} (-1)^k q^{k(3k-1)/2}$$

```
theorem tsum_qPochhammerInf_self {R : Type*} [CommRing R] [UniformSpace R] [IsUniformAddGroup R]
    [NonarchimedeanRing R] [CompleteSpace R] [T2Space R]
    {q : R} (hq : IsTopologicallyNilpotent q := by simp) :
    ∑' k : ℤ, k.negOnePow • q ^ pentagonal k = (q; q)_∞ :=
```

Wait...

Euler's Pentagonal Number Theorem

Using JTP, we can prove PNT by specializing it to $(a, b, q) := (q, q^2, q^3)$. If q is topologically nilpotent in a complete **non-archimedean** ring, then

$$(q; q)_\infty = \sum_{k \in \mathbb{Z}} (-1)^k q^{k(3k-1)/2}$$

```
theorem tsum_qPochhammerInf_self {R : Type*} [CommRing R] [UniformSpace R] [IsUniformAddGroup R]
    [NonarchimedeanRing R] [CompleteSpace R] [T2Space R]
    {q : R} (hq : IsTopologicallyNilpotent q := by simp) :
    ∑' k : ℤ, k.negOnePow • q ^ pentagonal k = (q; q)_∞ :=
```

Wait... Why only `NonarchimedeanRing`, not `StrongNonarchimedeanRing`?

Removing the strongness assumption

Theorem

Let R_1 be a ring and R_2 be a complete Hausdorff ***non-archimedean ring****. Let $y \in R_2$ be a topologically nilpotent element, and let $f\colon R_1 \to R_2$ be a ring homomorphism with bounded range. Then there is a canonical continuous ring homomorphism $\widetilde{f}\colon R_1[\![x]\!] \to R_2$ with $\widetilde{f}(x) = y$, where the topology on $R_1[\![x]\!]$ is the x-adic topology.*

For a topological ring R and a subset $S \subseteq R$, we say that S is **(topologically) bounded** if for every neighborhood U of 0, there is a neighborhood V of 0 such that $V \cdot S \subseteq U$.

Removing the strongness assumption

Theorem

Let R_1 be a ring and R_2 be a complete Hausdorff ***non-archimedean ring****. Let $y \in R_2$ be a topologically nilpotent element, and let $f\colon R_1 \to R_2$ be a ring homomorphism with bounded range. Then there is a canonical continuous ring homomorphism $\widetilde{f}\colon R_1[\![x]\!] \to R_2$ with $\widetilde{f}(x) = y$, where the topology on $R_1[\![x]\!]$ is the x-adic topology.*

For a topological ring R and a subset $S \subseteq R$, we say that S is **(topologically) bounded** if for every neighborhood U of 0, there is a neighborhood V of 0 such that $V \cdot S \subseteq U$.

$\mathbb{Z}[\![x]\!]$ is a strongly non-archimedean ring, so PNT holds in $\mathbb{Z}[\![x]\!]$. We can then map this identity to any non-archimedean ring R with a topologically nilpotent $q \in R$, obtaining PNT in R.

Removing the strongness assumption

```
variable {R} [CommRing R] [UniformSpace R] [NonarchimedeanRing R]
  [IsUniformAddGroup R] [CompleteSpace R] [T2Space R]

noncomputable def intEval (q : R) : ℤ⟦X⟧ →+* R := sEval (Int.castRingHom R) q

@[simp] theorem intEval_X {q : R} (hq : IsTopologicallyNilpotent q) : intEval q X = q := ...

@[fun_prop, simp] theorem intEval_continuous (q : R) : Continuous (intEval q) := ...

theorem eq_intEval (f : ℤ⟦X⟧ →+* R) (hf : Continuous f) : f = intEval (f X) := ...
```

`intEval_X` and continuity give existence; `eq_intEval` gives uniqueness among continuous ring homomorphisms.

Rogers–Ramanujan identities

There are many (more than 100!) proofs of the Rogers–Ramanujan identities. We follow a proof that applies Bailey's lemma twice. The following is a Bailey pair (with respect to $a = 1$):

$$\alpha_{1,n} = \begin{cases} 1 & n = 0 \\ (-1)^n q^{\binom{n}{2}}(1+q^n) & n > 0 \end{cases} \qquad \beta_{1,n} = \begin{cases} 1 & n = 0 \\ 0 & n > 0 \end{cases}$$

Applying the original form of Bailey's lemma produces a new Bailey pair

$$\alpha'_{1,n} = \begin{cases} 1 & n = 0 \\ q^{n^2} \cdot (-1)^n q^{\binom{n}{2}}(1+q^n) & n > 0 \end{cases} \qquad \beta'_{1,n} = \frac{1}{(q)_n}$$

Applying the second limiting version of Bailey's lemma to $(\alpha'_{1,n}, \beta'_{1,n})$ and then applying JTP to $(q^2;q^5)_\infty (q^3;q^5)_\infty (q^5;q^5)_\infty$ gives the first Rogers–Ramanujan identity.

Rogers–Ramanujan identities

```
theorem first_rogers_ramanujan_hasSum_strong {R} [CommRing R] [UniformSpace R]
  [IsUniformAddGroup R] [CompleteSpace R] [T2Space R] [StrongNonarchimedeanRing R]
  {q : R} (hq : IsTopologicallyNilpotent q) :
  HasSum (fun j ↦ q ^ j ^ 2 * bInv (q; q)_j) (bInv (q; q^5)_∞ * bInv (q^4; q^5)_∞) :=
```

Rogers–Ramanujan identities

```
theorem first_rogers_ramanujan_hasSum_strong {R} [CommRing R] [UniformSpace R]
  [IsUniformAddGroup R] [CompleteSpace R] [T2Space R] [StrongNonarchimedeanRing R]
  {q : R} (hq : IsTopologicallyNilpotent q) :
  HasSum (fun j ↦ q ^ j ^ 2 * bInv (q; q)_j) (bInv (q; q^5)_∞ * bInv (q^4; q^5)_∞) :=
```

Using the same trick as for PNT, we can upgrade the theorem:

```
theorem first_rogers_ramanujan {R} [CommRing R] [UniformSpace R]
  [IsUniformAddGroup R] [CompleteSpace R] [T2Space R] [NonarchimedeanRing R]
  {q : R} (hq : IsTopologicallyNilpotent q) :
  ∑' j, q ^ (j ^ 2) * bInv (q; q)_(j) = bInv (q; q^5)_∞ * bInv (q^4; q^5)_∞ :=
```

AxiomProver

AxiomProver

AxiomProver is an AI system currently under development at Axiom Math  . You can submit statements in natural or formal language to AxiomProver (as well as natural-language proofs, if needed), and it will produce formal proofs.

It was used to prove and automatically formalize several new results in Lean, including:

- JTP, PNT, and Jacobi's identity
- The theorem that the limit of a convergent sequence of topologically nilpotent elements in a strongly non-archimedean ring is also topologically nilpotent
- A counterexample to our conjecture on Huber pairs
- A denominator-cleared version of the q-Pfaff–Saalschütz identity (for Bailey's lemma)

AxiomProver: counterexample

Conjecture

Let R be a strongly non-archimedean ring. Let

$$R_0 := \{t \in R : \forall x \in R,\ x \text{ is topologically nilpotent} \Rightarrow tx \text{ is topologically nilpotent}\}$$

be the set of "nilpotent-stable" elements. Then R_0 is an open neighborhood of 0.

Informally, "all strongly non-archimedean rings arise from Huber pairs".

```
theorem nil_stable_not :
    ¬ ∀ (R : Type) [CommRing R] [TopologicalSpace R] [NonarchimedeanRing R],
    {x : R | ∀ q, IsTopologicallyNilpotent q → IsTopologicallyNilpotent (x * q)} ∈ nhds 0 := by
  sorry
```

So the formal input asked about non-archimedean rings. However, AxiomProver found a counterexample $R = \mathbb{Q}_p^{\mathbb{N}}$, which is in fact a strongly non-archimedean ring.

Future work

Possible future directions

- Ramanujan congruences for partition numbers, e.g., $p(5n + 4) \equiv 0 \pmod 5$.
- Connections between the existing $\eta(z)$ and $\theta(z)$ definitions in `mathlib` via JTP.
- Further applications of Bailey's lemma.
- p-adic or mod-p modular forms (cf. Serre, Katz, ...).

Thank you!

Happy formalization!

Paper and code:

- Kenny Lau, **L.**, Ken Ono, *Formalized q-series: The Rogers–Ramanujan Identities and Beyond*, `https://arxiv.org/abs/2607.01544`
- `https://github.com/AxiomMath/RogersRamanujan`
- `https://github.com/AxiomMath/RogersRamanujan-artifacts`